PARTIE I

Q1 a) Les valeurs propres de A sont strictement positives. Ainsi 0 n'est pas valeur propre de A et **A est inversible.**

L'équation $X \in \mathcal{M}_{n,1}(\mathbb{R})$ et $AX = B$ a alors une solution et une seule que nous noterons : R.

b) A est symétrique et réelle. Il existe une base orthonormale $(X_1, X_2, \dots, X_n)$ de $\mathcal{M}_{n,1}(\mathbb{R})$ constituée de vecteurs propres de A respectivement associés aux valeurs propres $\lambda_1, \lambda_2, \dots, \lambda_n$.

Soit X un élément non nul de E. $\exists (\alpha_1, \alpha_2, \dots, \alpha_n) \in \mathbb{R}^n$, $X = \sum_{i=1}^{n} \alpha_i X_i$

$AX = \sum_{i=1}^{n} \alpha_i A X_i = \sum_{i=1}^{n} \alpha_i \lambda_i X_i$

$$\phi(X) = {}^tXAX = \langle AX, X\rangle = \langle \sum_{i=1}^{n} \alpha_i \lambda_i X_i, \sum_{j=1}^{n} \alpha_j X_j \rangle \overset{(X_1, X_2, \dots, X_n) \text{ est une base orthonormale}}{=} \sum_{i=1}^{n} \alpha_i \lambda_i \alpha_i = \sum_{i=1}^{n} \lambda_i \alpha_i^2$$

X n'est pas nul. $\exists i_0 \in [\![1, n]\!]$, $\alpha_{i_0} \neq 0$. De plus : $\forall i \in [\![1, n]\!]$, $\lambda_i > 0$.

Par conséquent : $\phi(X) = \sum_{i=1}^{n} \lambda_i \alpha_i^2 \geqslant \lambda_{i_0} \alpha_{i_0}^2 > 0$

$\forall X \in E$, $X \neq 0 \Rightarrow \phi(X) > 0$.

c) $(X_1, X_2, \dots, X_n)$ est la base utilisée au-dessus.

Soit $X \in E$. $\exists (\alpha_1, \alpha_2, \dots, \alpha_n) \in \mathbb{R}^n$, $X = \sum_{i=1}^{n} \alpha_i X_i$. $\|X\|^2 = \sum_{i=1}^{n} \alpha_i^2$.

$\phi(X) = \langle AX, X\rangle = \sum_{i=1}^{n} \lambda_i \alpha_i^2$. Posons $\lambda = \min_{1 \leqslant i \leqslant n} \lambda_i$ et $\mu = \max_{1 \leqslant i \leqslant n} \lambda_i$.

$\forall i \in [\![1, n]\!]$, $\lambda \alpha_i^2 \leqslant \lambda_i \alpha_i^2 \leqslant \mu \alpha_i^2$; $\lambda \sum_{i=1}^{n} \alpha_i^2 \leqslant \sum_{i=1}^{n} \lambda_i \alpha_i^2 \leqslant \mu \sum_{i=1}^{n} \alpha_i^2$

Donc $\lambda \|X\|^2 \leqslant \phi(X) \leqslant \mu \|X\|^2$.

$\exists (\lambda, \mu) \in \mathbb{R}_+^{*2}$, $\forall X \in E$, $\lambda \|X\|^2 \leqslant \phi(X) \leqslant \mu \|X\|^2$.

d) Soit X un élément non nul de E.

Soit Y un élément de E.

$\langle AX, Y\rangle = 0 \iff Y$ est orthogonal à $AX \iff Y \in [\mathrm{Vect}(AX)]^{\perp}$

Ainsi **l'ensemble des vecteurs de E, conjugués de X par rapport à ϕ est le sous-espace vectoriel $[\mathrm{Vect}(AX)]^{\perp}$.**

X n'est pas nul et A est inversible donc AX n'est pas nul ; ainsi $\mathrm{Vect}(AX)$ est une droite vectorielle. Par conséquent : $\dim[\mathrm{Vect}(AX)]^{\perp} = n-1$.

(Q2) $\forall X=(x_1,x_2,\dots,x_n)\in E=\mathbb{R}^n,\ F(X)={}^tXAX-2\langle B,X\rangle$

Supposons que $A=(a_{ij})$ et $B=(b_1,b_2,\dots,b_n)$

$X=(x_1,x_2,\dots,x_n)\in E,\quad F(X)=\sum_{i=1}^n x_i\left(\sum_{j=1}^n a_{ij}x_j\right)-2\sum_{i=1}^n b_ix_i=\sum_{i=1}^n\sum_{j=1}^n a_{ij}x_ix_j-2\sum_{i=1}^n b_ix_i$

Transformons le double Σ.

$$\sum_{i=1}^n\sum_{j=1}^n a_{ij}x_ix_j=\sum_{1\le i<j\le n}a_{ij}x_ix_j+\sum_{i=1}^n a_{ii}x_i^2+\sum_{1\le j<i\le n}a_{ij}x_ix_j$$

$$=\sum_{1\le i<j\le n}a_{ij}x_ix_j+\sum_{i=1}^n a_{ii}x_i^2+\underset{i\leftrightarrow j}{\sum_{1\le i<j\le n}}a_{ji}x_jx_i\ \overset{a_{ij}=a_{ji}}{=}\ \sum_{i=1}^n a_{ii}x_i^2+2\sum_{1\le i<j\le n}a_{ij}x_ix_j$$

$$\forall X=(x_1,x_2,\dots,x_n)\in\mathbb{R}^n,\ F(X)=\sum_{i=1}^n a_{ii}x_i^2+2\sum_{1\le i<j\le n}a_{ij}x_ix_j-2\sum_{i=1}^n b_ix_i.$$

(*) A rédiger et à savoir faire

F étant une fonction polynôme, F est en particulier de classe $\mathcal{C}^1$ sur $\mathbb{R}^n$.

$$\forall k\in[\![1,n]\!],\ \forall X=(x_1,\dots,x_n)\in\mathbb{R}^n,\ \frac{\partial F}{\partial x_k}(X)\overset{(*)}{=}a_{kk}2x_k+2\sum_{i=1}^{k-1}a_{ik}x_i\times 1+2\sum_{j=k+1}^n a_{kj}1\times x_j-2b_k.$$

$$\forall k\in[\![1,n]\!],\ \forall X=(x_1,\dots,x_n)\in\mathbb{R}^n,\ \frac{\partial F}{\partial x_k}(X)=2a_{kk}x_k+2\sum_{i=1}^{k-1}\underbrace{a_{ki}}_{a_{ik}=a_{ki}}x_i+2\underset{j\to i}{\sum_{i=k+1}^n}a_{ki}x_i-2b_k.$$

$$\forall k\in[\![1,n]\!],\ \forall X=(x_1,x_2,\dots,x_n)\in\mathbb{R}^n,\ \frac{\partial F}{\partial x_k}(X)=2\left[\sum_{i=1}^n a_{ki}x_i-b_k\right].$$

Ainsi $\forall X\in E,\ \mathrm{grad}\,F(X)=2(AX-B)=G(X)$.

En particulier $\mathrm{grad}\,F(R)=2(AR-B)=0$; $G(R)=0$.

Po (*) Pour bien comprendre il suffit d'écrire :

$$\sum_{1\le i<j\le n}a_{ij}x_ix_j=\sum_{1\le i<j<k}a_{ij}x_ix_j+\sum_{i=1}^{k-1}a_{ik}x_ix_k+\sum_{\substack{1\le i<j\le n\\ \text{et } j>k}}a_{ij}x_ix_j$$

Q3) $(X,H)\in E^2$.

$F(X+H) = \phi(X+H) - 2\langle B, X+H\rangle = {}^t(X+H)A(X+H) - 2\langle B,X\rangle - 2\langle B,H\rangle$

$F(X+H) = {}^tXAX + {}^tXAH + {}^tHAX + {}^tHAH - 2\langle B,X\rangle - 2\langle B,H\rangle$. Notons que : ${}^tXAH = {}^tHAX = \langle AX, H\rangle$.

$F(X+H) = {}^tXAX - 2\langle B,X\rangle + 2\langle AX,H\rangle - 2\langle B,H\rangle + {}^tHAH = F(X) + 2\langle AX-B, H\rangle + \phi(H)$.

Ainsi : $\underline{\underline{\forall (X,H)\in E^2,\quad F(X+H) = F(X) + 2\langle G(X), H\rangle + \phi(H).}}$

$\forall H\in E$, $F(R+H) = F(R) + 2\langle G(R), H\rangle + \phi(H) \overset{G(R)=0}{=} F(R) + \phi(H)$.

Or $\forall H\in E$, $\phi(H)\geqslant 0$; mieux : $\forall H\in E-\{0\}$, $\phi(H)>0$.

Ainsi $\forall H\in E-\{0\}$, $F(R+H) = F(R)+\phi(H) > F(R)$. $\underline{\underline{\forall H\in E-\{0\},\ F(R+H)>F(R).}}$

F présente donc un minimum strict en R.

Q4) a) Soit $X\in E-\{R\}$. Soit $\rho\in\mathbb{R}$.

$F(X-\rho G(X)) = F(X) + \langle G(X), -\rho G(X)\rangle + \phi(-\rho G(X))$

$F(X-\rho G(X)) = F(X) - \rho\|G(X)\|^2 + (-\rho)^2\phi(G(X))$

$F(X-\rho G(X)) = \phi(G(X))\rho^2 - \rho\|G(X)\|^2 + F(X)$.

$G(X) = 2(AX-B) \neq 0$ car $X\neq R$

donc $F(X-\rho G(X)) = \phi(G(X))\left[\rho^2 - \dfrac{\|G(X)\|^2}{\phi(G(X))} + \dfrac{F(X)}{\phi(G(X))}\right]$.

$F(X-\rho G(X)) = \phi(G(X))\left[\left(\rho - \dfrac{\|G(X)\|^2}{2\phi(G(X))}\right)^2 + \dfrac{F(X)}{\phi(G(X))} - \dfrac{\|G(X)\|^4}{(2\phi(G(X)))^2}\right]$. Notons que $\phi(G(X))>0$.

Ainsi $F(X-\rho G(X))$ est minimum pour $\rho = \dfrac{\|G(X)\|^2}{2\phi(G(X))}$

Pour $\rho = \dfrac{\|G(X)\|^2}{2\phi(G(X))}$, $F(X-\rho G(X)) = \phi(G(X))\left[\dfrac{F(X)}{\phi(G(X))} - \dfrac{\|G(X)\|^4}{(2\phi(G(X)))^2}\right] = F(X) - \dfrac{\|G(X)\|^4}{4\phi(G(X))}$.

Si $X\in E-\{R\}$, $F(X-\rho G(X))$ est minimum pour $\rho = \dfrac{\|G(X)\|^2}{2\phi(G(X))}$; ce minimum est $F(X) - \dfrac{\|G(X)\|^4}{4\phi(G(X))}$

b) Ici $\rho = \dfrac{\|G(X)\|^2}{2\,\phi(G(X))}$ $(X \in E - \{0\})$. Posons $Y = X - \rho\, G(X)$.

$\langle G(X), G(Y)\rangle = \langle G(X), 2(AY-B)\rangle = \langle G(X), 2(AY-AX) + 2(AX-B)\rangle = \langle G(X), 2A(Y-X) + G(X)\rangle$

$\langle G(X), G(Y)\rangle = \langle G(X), 2A(-\rho\, G(X)) + G(X)\rangle = \langle G(X), G(X)\rangle - 2\rho \langle A\,G(X), G(X)\rangle$

$\langle G(X), G(Y)\rangle = \|G(X)\|^2 - 2\rho\, \phi(G(X)) = \|G(X)\|^2 - 2 \times \dfrac{\|G(X)\|^2}{2\,\phi(G(X))}\, \phi(G(X)) = 0$.

Ainsi si $X \in E - \{0\}$, si $\rho = \dfrac{\|G(X)\|^2}{2\,\phi(G(X))}$ et si $Y = X - \rho\, G(X)$ alors $\langle G(X), G(Y)\rangle = 0$.

Q 5 a) Soit $k \in \mathbb{N}$.

* Supposons $X_k \neq R$. Alors $X_{k+1} = X_k - \rho_k\, G(X_k)$ avec $\rho_k = \dfrac{\|G(X_k)\|^2}{2\,\phi(G(X_k))}$. Donc d'après Q4 a) : $F(X_{k+1}) \leq F(X_k)$

* Supposons $X_k = R$. Alors $X_{k+1} = R$. Donc $F(X_{k+1}) = F(X_k)$ et ainsi $F(X_{k+1}) \leq F(X_k)$!

Finalement : $\forall k \in \mathbb{N}$, $F(X_{k+1}) \leq F(X_k)$; $(F(X_k))_{k \geq 0}$ est décroissante.

Rappelons que Q3 nous a montré que : $\forall X \in E$, $F(X) \geq F(R)$.

Par conséquent : $(F(X_k))_{k\geq 0}$ est décroissante et minorée par $F(R)$; $(F(X_k))_{k \geq 0}$ converge.

$(F(X_{k+1}) - F(X_k))_{k\geq 0}$ est alors convergente comme différence de deux suites convergentes.

Remarque .. $\lim\limits_{k \to +\infty} (F(X_{k+1}) - F(X_k)) = 0$ ..

b) Soit $k \in \mathbb{N}$. Supposons $X_k \neq R$. $X_{k+1} = X_k - \rho_k\, G(X_k)$ avec $\rho_k = \dfrac{\|G(X_k)\|^2}{2\,\phi(G(X_k))}$.

Ainsi d'après Q4 a) $F(X_{k+1}) = F(X_k - \rho_k\, G(X_k)) = F(X_k) - \dfrac{\|G(X_k)\|^4}{4\,\phi(G(X_k))}$.

Donc $0 \leq \dfrac{\|G(X_k)\|^4}{4\,\phi(G(X_k))} = F(X_k) - F(X_{k+1})$

$$0 \leq \|G(X_k)\|^4 = 4\,\phi(G(X_k))\,(F(X_k) - F(X_{k+1})) \underset{\substack{F(X_k)-F(X_{k+1}) \geq 0 \\ \text{Q1 c)}}}{\leq} 4\mu\, \|G(X_k)\|^2\, (F(X_k) - F(X_{k+1}))$$

Or $X_k \neq R$ donc $G(X_k) \neq 0$; $\|G(X_k)\| \neq 0$.

Nous pouvons alors écrire : $0 \leq \|G(X_k)\|^2 \leq 4\mu\,(F(X_k) - F(X_{k+1}))$.

Il ne reste plus qu'à remarquer que cet encadrement vaut encore si $X_k = R$

En effet si $X_k = R$: $X_{k+1} = R$. Ainsi $\|G(X_k)\|^2 = 0$ et $F(X_k) - F(X_{k+1}) = 0$.

Finalement : $\forall k \in \mathbb{N}$, $0 \leq \|G(X_k)\|^2 \leq 4\mu\,(F(X_k) - F(X_{k+1}))$.

Comme $\lim\limits_{k \mapsto +\infty} (F(X_{k+1}) - F(X_k)) = 0$ il vient par encadrement : $\lim\limits_{k \mapsto +\infty} \|G(X_k)\|^2 = 0$

Ainsi $\lim\limits_{k \mapsto +\infty} \|G(X_k)\| = 0$ et $(G(X_k))_{k \geq 0}$ converge vers le vecteur nul.

$\forall k \in \mathbb{N}$, $G(X_k) = 2(AX_k - B)$; $\forall k \in \mathbb{N}$, $AX_k = \frac{1}{2} G(X_k) + B$

$\forall k \in \mathbb{N}$, $X_k = \frac{1}{2} A^{-1} G(X_k) + A^{-1} B = \frac{1}{2} A^{-1} G(X_k) + R$

$\lim\limits_{k \mapsto +\infty} G(X_k) = 0$ donc $\lim\limits_{k \mapsto +\infty} (A^{-1} G(X_k)) = A^{-1} 0 = 0$. Ainsi $\lim\limits_{k \mapsto +\infty} X_k = R$.

(Q6) Le texte est sans malice... ou presque.

X_0 est arbitraire. On passe de X_k à X_{k+1} de connaissant X_k, on calcule $G(X_k)$. Si $G(X_k) = 0$ (identique à $\|G(X_k)\| < \varepsilon$ où ε est donné par l'utilisateur) on s'arrête. Dans le cas contraire on calcule ρ_k et on pose $X_{k+1} = X_k - \rho_k G(X_k)$.

⚠ 23-Oct-1999 - 18h36 - Je viens de toucher mon nouveau Waterman. Il faut que je m'habitue !

Dans le programme qui suit j'ai volontairement réduit le programme principal à sa plus simple expression. D'où l'utilisation systématique de procédures (et d'une fonction)

```
program GradienConjugue;
(*essai avec 1 1 0  1 2 1  0 1 2*)
uses crt,printer;

const DimMax=10;epsilon=1e-7;
type Matrice=array[1..DimMax,1..DimMax] of real;
     Vecteur=array[1..DimMax] of real;

var  k,n:integer;rho,test:real;a:Matrice;b,x,t,aux:vecteur;

(*****************************************************************************)

procedure Ecrit_Matrice(n:integer;A:matrice);
var i,j:integer;
begin
writeln;
for i:=1 to n do
    begin
    for j:=1 to n do write(A[i,j]:15:10);
    writeln;
    end;
end;

(*****************************************************************************)

Procedure EcritVecteur(n:integer;v:vecteur);
var i:integer;
begin
for i:=1 to n do writeln('La composante numéro ',i,' est : ',v[i]);
end;

(*****************************************************************************)

Procedure Entre_Matrice(n:integer;var a:Matrice);
var i,j:integer;

begin
for j:=1 to n do
for i:=1 to n do
    begin
    write('Donnez le coefficient de la ligne numéro ',i);
    write(' et de la colonne numéro ',j,' : ');readln(a[i,j]);
    end;
end;

(*****************************************************************************)

procedure Entre_vecteur(n:integer;var u:vecteur);

var i:integer;
begin
for i:=1 to n do
    begin
    write('Donnez la composante numéro ',i,'  ');readln(u[i]);
    end;
end;

(*****************************************************************************)

procedure MultMatVect(n:integer;a:Matrice;v:Vecteur;var res:Vecteur);

var i,j:integer;s:real;
```

```
begin

for i:=1 to n do
    begin
    s:=0;
    for j:=1 to n do s:=s+a[i,j]* v[j];
    res[i]:=s;
    end;
end;

(***************************************************************************)

procedure MultScalVect(n:integer;v:vecteur;x:real;var res:vecteur);

var i:integer;
begin
for i:=1 to n do res[i]:=x*v[i];
end;

(***************************************************************************)

function ProdScal(n:integer;u,v:vecteur):real;

var i:integer;s:real;
begin
s:=0;for i:=1 to n do s:=s+u[i]*v[i];
ProdScal:=s;
end;

(***************************************************************************)

procedure AddVect(n:integer;u,v:vecteur;var res:vecteur);

var i:integer;

begin
for i:=1 to n do res[i]:=u[i]+v[i];
end;

(***************************************************************************)

begin
clrscr;
write('Donnez la dimension n. n=');readln(n);
writeln;writeln('Introduction de la matrice');
Entre_Matrice(n,a);
writeln;writeln('Introduction du vecteur');
Entre_Vecteur(n,b);

for k:=1 to n do x[k]:=1;k:=0;

MultMatVect(n,a,x,t);          (*Calcul de AX*)
MultScalVect(n,t,2,t);         (*Calcul de 2AX*)
MultScalVect(n,b,-2,aux);      (*Calcul de -2B*)
AddVect(n,t,aux,t);            (*Calcul de G(X)=2AX-2B*)
test:=ProdScal(n,t,t);         (*Cacul de ||G(X)||²*)

while sqrt(test)>=epsilon do
      begin
      MultMatVect(n,a,t,aux);              (*Calcul de AG(X)*)
      rho:=test/2/ProdScal(n,aux,t);       (*Calcul de rho*)
      MultScalVect(n,t,-rho,t);            (*Calcul de -rho G(X)*)
      k:=k+1;
      AddVect(n,t,x,x);                    (*Calcul de X-rho G(X)*)
```

```
        (*Début du calcul du nouveau terme*)

        MultMatVect(n,a,x,t);           (*Calcul de AX*)
        MultScalVect(n,t,2,t);          (*Calcul de 2AX*)
        MultScalVect(n,b,-2,aux);       (*Calcul de -2B*)
        AddVect(n,t,aux,t);             (*Calcul de G(X)=2AX-2B*)
        test:=ProdScal(n,t,t);          (*Cacul de ||G(X)||²*)

        end;

    writeln;
    writeln('La solution est sensiblement :');writeln;
    EcritVecteur(n,x);writeln;
    writeln;writeln('On a calculé ',k,' termes.');
    writeln;
    end.
```

PARTIE II

Q1 $R_0 = 0 \iff AX_0 - B = 0 \iff X_0 = R$

donc $R_0 = 0$ si et seulement si le premier terme de la suite est la solution !

Q2 $R_0 \neq 0$. donc $D_0 = R_0 \neq 0$; $\phi(D_0) > 0$; $\langle AD_0, D_0\rangle > 0$; ainsi $\alpha_0 = \dfrac{\langle R_0, D_0\rangle}{\langle AD_0, D_0\rangle}$ a un sens. $X_1 = X_0 - \alpha_0 D_0$, $R_1 = AX_1 - B$, $\beta_1 = \dfrac{\langle R_1, R_1\rangle}{\langle R_0, R_0\rangle}$

et $D_1 = R_1 + \beta_0 D_0$ sont alors définis.

<u>Si R_0 n'est pas nul : X_1, R_1 et D_1 existent.</u>

$$\langle R_1, R_0\rangle = \langle AX_1 - B, R_0\rangle = \langle A(X_0 - \alpha_0 D_0) - B, R_0\rangle = \langle \overbrace{AX_0 - B}^{R_0} - \alpha_0 AD_0, R_0\rangle$$

$$\langle R_1, R_0\rangle = \langle R_0 - \alpha_0 AD_0, R_0\rangle = \langle R_0, R_0\rangle - \alpha_0\langle AD_0, R_0\rangle \underset{R_0 = D_0}{=} \langle R_0, D_0\rangle - \frac{\langle R_0, D_0\rangle}{\langle AD_0, D_0\rangle}\langle AD_0, D_0\rangle$$

Ainsi $\underline{\underline{\langle R_1, R_0\rangle = 0}}$.

$$\langle AD_1, D_0\rangle \underset{A \text{ est symétrique}}{=} \langle D_1, AD_0\rangle = \langle R_1 + \beta_0 D_0, AD_0\rangle = \langle R_1, AD_0\rangle + \beta_0\langle D_0, AD_0\rangle.$$

$$R_1 = AX_1 - B = A(X_0 - \alpha_0 D_0) - B = AX_0 - B - \alpha_0 AD_0 = R_0 - \alpha_0 AD_0 \; ; \; AD_0 = \frac{1}{\alpha_0}(R_0 - R_1)$$

donc $\langle AD_1, D_0\rangle = \langle R_1, \frac{1}{\alpha_0}(R_0 - R_1)\rangle + \beta_0 \frac{1}{\alpha_0}\langle R_0, D_0\rangle$ car $\langle D_0, AD_0\rangle = \frac{1}{\alpha_0}\langle R_0, D_0\rangle$

$$\langle AD_1, D_0\rangle = \frac{1}{\alpha_0}\Big[\langle R_1, R_0\rangle \underbrace{- \langle R_1, R_1\rangle + \frac{\langle R_1, R_1\rangle}{\langle R_0, R_0\rangle}\langle R_0, D_0\rangle}_{= 0 \text{ car } R_0 = D_0}\Big] = \frac{1}{\alpha_0}\big[\langle R_1, R_0\rangle\big] \underset{\text{voir plus haut.}}{=} 0$$

Ainsi $\underline{\underline{\langle R_0, R_1\rangle = 0 \text{ et } \langle AD_1, D_0\rangle = 0.}}$

Q3 a) Montrons par récurrence que pour tout $i \in [\![0,k]\!]$, $D_0, D_1, \dots, D_i$ existent et $\text{Vect}(D_0, D_1, \dots, D_i) = \text{Vect}(R_0, R_1, \dots, R_i)$.

→ C'est clair pour $i=0$ car $D_0 = R_0$

→ Supposons la propriété vraie pour $i \in [\![0,k-1]\!]$ et montrons la pour $i+1$.

D_{i+1} est bien défini car β_i est défini puisque $\langle R_i, R_i\rangle = \|R_i\|^2 \neq 0$.

$D_{i+1} = R_{i+1} + \beta_i D_i \in \text{Vect}(R_0, R_1, \dots, R_i, R_{i+1})$ car $D_i \in \text{Vect}(R_0, R_1, \dots, R_i)$

$D_{i+1} \in \text{Vect}(R_0, R_1, \dots, R_i, R_{i+1})$ et $\text{Vect}(D_0, D_1, \dots, D_i) \overset{H.R.}{=} \text{Vect}(R_0, R_1, \dots, R_i)$; ainsi :

$\text{Vect}(D_0, D_1, \dots, D_i, D_{i+1}) \subset \text{Vect}(R_0, R_1, \dots, R_i, R_{i+1})$. Montrons alors que ces deux sous-espaces vectoriels sont égaux.

Observons que : $R_{i+1} = D_{i+1} - \beta_i D_i \in \text{Vect}(D_i, D_{i+1}) \subset \text{Vect}(D_0, D_1, \dots, D_i, D_{i+1})$.

$R_{i+1} \in \text{Vect}(D_0, D_1, \dots, D_i, D_{i+1})$ et $\text{Vect}(R_0, R_1, \dots, R_i) = \text{Vect}(D_0, D_1, \dots, D_i)$ donnent alors

$\text{Vect}(R_0, R_1, \dots, R_i, R_{i+1}) \subset \text{Vect}(D_0, D_1, \dots, D_i, D_{i+1})$.

Ainsi $\text{Vect}(D_0, D_1, \dots, D_i, D_{i+1}) = \text{Vect}(R_0, R_1, \dots, R_i, R_{i+1})$ ce qui achève la récurrence.

<u>Pour tout $i \in [\![0,k]\!]$, $D_0, D_1, \dots, D_i$ existent et $\text{Vect}(D_0, D_1, \dots, D_i) = \text{Vect}(R_0, R_1, \dots, R_i)$.</u>

$(R_0, R_1, \dots, R_k)$ est une famille de vecteurs non nuls deux à deux orthogonaux donc c'est une famille libre. $(R_0, R_1, \dots, R_k)$ est donc une base de $V_k = \text{Vect}(R_0, R_1, \dots, R_k)$.

Ainsi $\dim V_k = k+1$. $D_0, D_1, \dots, D_k$ est alors une famille génératrice de $k+1$ vecteurs du sous-espace vectoriel V_k qui est de dimension $k+1$. $(D_0, D_1, \dots, D_k)$ est alors une base de V_k. $\underline{\underline{(D_0, D_1, \dots, D_k) \text{ est une famille libre de } E.}}$

b) Soit $(i,j) \in [\![0,k]\!]^2$ tel que : $j > i$

R_j est orthogonal à $R_0, R_1, \dots, R_i$ donc $R_j \in [\text{Vect}(R_0, R_1, \dots, R_i)]^\perp$

Ainsi $R_j \in [\text{Vect}(D_0, D_1, \dots, D_i)]^\perp$ et R_j est alors orthogonal à D_i.

$\forall (i,j) \in [\![0,k]\!]^2, \quad j > i \Rightarrow \langle D_i, R_j\rangle = 0.$

c) Montrons par récurrence que, pour tout $j \in [\![1,k]\!]$. $\forall i \in [\![0,j-1]\!]$, $\langle AD_i, D_j \rangle = 0$.

Si $k=0$ c'est clair. Supposons $k \geq 1$.

* $j=1$. Il s'agit simplement de montrer que : $\langle AD_0, D_1 \rangle = 0$. C'est clair d'après Q2.
* Supposons la propriété vraie pour $j \in [\![1,k-1]\!]$ et montrons la pour $j+1$.

(H) $\forall i \in [\![0,j-1]\!]$, $\langle AD_i, D_j \rangle = 0$ (C) $\forall i \in [\![0,j]\!]$, $\langle AD_i, D_{j+1} \rangle = 0$.

• Soit $i \in [\![0,j-1]\!]$.

$$\langle AD_i, D_{j+1} \rangle = \langle AD_i, R_{j+1} + \beta_j D_j \rangle = \langle AD_i, R_{j+1} \rangle + \beta_j \langle AD_i, D_j \rangle \overset{H.R}{=} \langle AD_i, R_{j+1} \rangle.$$

Montrons alors que $\langle AD_i, R_{j+1} \rangle = 0$.

$R_{i+1} = AX_{i+1} - B = A(X_i - \alpha_i D_i) - B = (AX_i - B) - \alpha_i AD_i = R_i - \alpha_i AD_i$; $\alpha_i AD_i = R_i - R_{i+1}$.

Supposons $\alpha_i = \dfrac{\langle R_i, D_i \rangle}{\langle AD_i, D_i \rangle} = 0$; $\langle R_i, D_i \rangle = 0$; R_i est orthogonal à D_i.

Or R_i est orthogonal à $D_0, D_1, \dots, D_{i-1}$ d'après b. Ainsi R_i est orthogonal à $D_0, D_1, \dots, D_i$

$R_i \in [\mathrm{Vect}(D_0, D_1, \dots, D_i)]^\perp = [\mathrm{Vect}(R_0, R_1, \dots, R_i)]^\perp$; R_i est alors orthogonal à R_i ; $R_i = 0$!

Ainsi : $\alpha_i \neq 0$ et $AD_i = \frac{1}{\alpha_i}(R_i - R_{i+1})$.

Donc $\langle AD_i, D_{j+1} \rangle = \langle AD_i, R_{j+1} \rangle = \frac{1}{\alpha_i} \langle R_i, R_{j+1} \rangle - \frac{1}{\alpha_i} \langle R_{i+1}, R_{j+1} \rangle \overset{i<j+1 \text{ et } i+1<j+1}{=} 0$

• Supposons $i=j$ et montrons que $\langle AD_i, D_{j+1} \rangle = 0$.

Soit à montrer que : $\langle AD_j, D_{j+1} \rangle = 0$.

$\langle AD_j, D_{j+1} \rangle = \langle AD_j, R_{j+1} + \beta_j D_j \rangle = \langle AD_j, R_{j+1} \rangle + \beta_j \langle AD_j, D_j \rangle$. Or $AD_j = \frac{1}{\alpha_j}(R_j - R_{j+1})$

$$\langle AD_j, D_{j+1} \rangle = \left\langle \frac{1}{\alpha_j}(R_j - R_{j+1}), R_{j+1} \right\rangle + \beta_j \langle AD_j, D_j \rangle$$

$$= \frac{1}{\alpha_j}\left[\langle R_j, R_{j+1} \rangle - \|R_{j+1}\|^2 + \beta_j \alpha_j \langle AD_j, D_j \rangle\right]$$

$$= \frac{1}{\alpha_j}\Big[\underbrace{\langle R_j, R_{j+1} \rangle}_{=0} - \|R_{j+1}\|^2 + \frac{\langle R_{j+1}, R_{j+1} \rangle}{\langle R_j, R_j \rangle} \, \frac{\langle R_j, D_j \rangle}{\langle AD_j, D_j \rangle} \langle AD_j, D_j \rangle$$

$$= \frac{1}{\alpha_j}\left[-\|R_{j+1}\|^2 + \frac{\|R_{j+1}\|^2}{\|R_j\|^2} \langle R_j, D_j \rangle\right]$$

Or $\langle R_j, D_j \rangle = \langle R_j, R_j + \beta_{j-1} D_{j-1} \rangle = \|R_j\|^2 - \beta_{j-1} \langle R_j, D_{j-1} \rangle \overset{b)}{=} \|R_j\|^2$

Ainsi $\langle AD_j, D_{j+1} \rangle = \frac{1}{\alpha_j}\left[-\|R_{j+1}\|^2 + \frac{\|R_{j+1}\|^2}{\|R_j\|^2} \|R_j\|^2\right] = 0$ et la récurrence s'achève.

Finalement : $\underline{\underline{\forall j \in [\![1,k]\!],\ \forall i \in [\![0,j-1]\!],\ \langle AD_i, D_j\rangle = 0}}$.

Soit $(i,j) \in [\![0,k]\!]^2$ tel que $i \neq j$.

1er cas.. $i<j$. Alors $j \in [\![1,k]\!]$ et $i \in [\![0,j-1]\!]$ donc $\langle AD_i, D_j\rangle = 0$

2e cas.. $i>j$ Alors : $\langle AD_i, D_j\rangle = \langle D_i, AD_j\rangle = \langle AD_j, D_i\rangle \underset{\substack{\uparrow \\ j<i + 1^{er} \text{ cas !}}}{=} 0$

Finalement : $\underline{\underline{\forall (i,j) \in [\![0,k]\!]^2,\ i \neq j \Rightarrow \langle AD_i, D_j\rangle = 0}}$.

d) Pour que R_{k+1} existe il suffit que α_k existe donc que $\langle AD_k, D_k\rangle \neq 0$.

Or $\langle AD_k, D_k\rangle = \phi(D_k)$. D_k n'est pas nul car $(D_0, D_1, \dots, D_k)$ est libre donc $\phi(D_k) > 0$, ainsi $\langle AD_k, D_k\rangle$ n'est pas nul et R_{k+1} existe.

Montrons que R_{k+1} est orthogonal à $R_0, R_1, \dots, R_k$

$R_{k+1} = AX_{k+1} - B = A(X_k - \alpha_k D_k) - B = (AX_k - B) - \alpha_k AD_k = R_k - \alpha_k AD_k$

AD_k est orthogonal à $D_0, D_1, \dots, D_{k-1}$ donc à $R_0, R_1, \dots, R_{k-1}$ car $\text{Vect}(D_0, \dots, D_{k-1}) = \text{Vect}(R_0, R_1, \dots, R_{k-1})$.

R_k est également orthogonal à $R_0, R_1, \dots, R_{k-1}$ donc $R_{k+1} = R_k - \alpha_k AD_k$ est orthogonal à $R_0, R_1, \dots, R_{k-1}$.

Montrons pour finir que R_{k+1} est orthogonal à R_k.

$\langle R_{k+1}, D_k\rangle = \langle R_k - \alpha_k AD_k, D_k\rangle = \|R_k\|^2 - \alpha_k \langle AD_k, D_k\rangle = \langle R_k, R_k\rangle - \langle R_k, D_k\rangle$

$\langle R_{k+1}, D_k\rangle = \langle R_k, R_k - D_k\rangle$

Si $k=0$: $\langle R_{k+1}, D_k\rangle = \langle R_k, R_k - D_k\rangle = 0$ car $R_k - D_k = R_0 - D_0 = 0$.

Si $k \neq 0$: $\langle R_{k+1}, D_k\rangle = \langle R_k, R_k - D_k\rangle = \langle R_k, -\beta_{k-1} D_{k-1}\rangle = -\beta_{k-1}\langle R_k, D_{k-1}\rangle = 0$

Donc R_{k+1} est orthogonal à D_k et à $R_0, R_1, \dots, R_{k-1}$; ainsi R_{k+1} est orthogonal à $D_0, D_1, \dots, D_k$ car $\text{Vect}(R_0, R_1, \dots, R_{k-1}) = \text{Vect}(D_0, D_1, \dots, D_{k-1})$.

Ainsi $R_{k+1} \in [\text{Vect}(D_0, D_1, \dots, D_k)]^\perp = [\text{Vect}(R_0, R_1, \dots, R_k)]^\perp$; R_{k+1} est orthogonal à R_k.

Finalement $\underline{\underline{R_{k+1} \text{ existe et } R_{k+1} \text{ est orthogonal à } R_0, R_1, \dots, R_k}}$.

Q4 a) Si I est vide : $R_0 = 0$ et d'après Q1 $X_0 = R$

b) si $i \in I$, $(R_0, R_1, \dots, R_i)$ est libre car les vecteurs de cette famille sont non nuls et deux à deux orthogonaux ; par conséquent $i+1 \leq n$; $i \leq n-1$.

Dans cette question I est alors une partie non vide et majorée de $\mathbb{N}$ donc I possède un plus grand élément k. $k \leq n-1$.

D'après Q3 d) R_{k+1} existe et est orthogonal à $R_0, R_1, \dots, R_k$.

Si R_{k+1} n'est pas nul alors $k+1 \in I$!! ceci est impossible ($k = \max I$) donc $R_{k+1} = 0$. On a alors $0 = R_{k+1} = AX_{k+1} - B$; $AX_{k+1} = B$; $X_{k+1} = R$.

Ainsi $R_{k+1} = 0$ et $X_{k+1} = R$ si $I \neq \emptyset$ et si $k = \max I$.

c) Dans les deux cas ($I = \emptyset$ et $I \neq \emptyset$) on obtient la solution en au plus n itérations.

Q5 Ici on calcule $R_0, R_1, \dots, R_k$ jusqu'à obtenir $R_k = 0$ (ie $\|R_k\| < \varepsilon$) et la solution est alors X_k ... ou presque.

Pour changer nous n'utiliserons que les procédures permettant de lire ou d'écrire une matrice et un vecteur ... et la fonction produit scalaire.

Pour simplifier on peut remarquer que • $R_{k+1} = R_k - d_k A D_k$ • $\langle R_k, D_k \rangle = \|R_k\|^2$

```
program GradienConjugue;
(*essai avec 1 1 0  1 2 1  0 1 2*)
uses crt,printer;

const DimMax=10;epsilon=1e-9;
type Matrice=array[1..DimMax,1..DimMax] of real;
     Vecteur=array[1..DimMax] of real;

var  k,n,j,i:integer;epsi,norme,alpha,s,beta:real;
     a:Matrice;b,x,r,d,t:vecteur;

(***************************************************************************)

procedure Ecrit_Matrice(n:integer;A:matrice);
var i,j:integer;
begin
writeln;
for i:=1 to n do
    begin
    for j:=1 to n do write(A[i,j]:15:10);
    writeln;
    end;
end;

(***************************************************************************)

Procedure EcritVecteur(n:integer;v:vecteur);
var i:integer;
begin
for i:=1 to n do writeln('La composante numéro ',i,' est : ',v[i]);
end;

(***************************************************************************)

Procedure Entre_Matrice(n:integer;var a:Matrice);
var i,j:integer;

begin
for j:=1 to n do
for i:=1 to n do
    begin
    write('Donnez le coefficient de la ligne numéro ',i);
    write(' et de la colonne numéro ',j,' : ');readln(a[i,j]);
    end;
end;

(***************************************************************************)

procedure Entre_vecteur(n:integer;var u:vecteur);

var i:integer;
begin
for i:=1 to n do
    begin
    write('Donnez la composante numéro ',i,'  ');readln(u[i]);
    end;
end;

(***************************************************************************)

function ProdScal(n:integer;u,v:vecteur):real;
```

```
var i:integer;s:real;
begin
s:=0;for i:=1 to n do s:=s+u[i]*v[i];
ProdScal:=s;
end;

(***************************************************************************)

begin
clrscr;
write('Donnez la dimension n. n=');readln(n);
writeln;writeln('Introduction de la matrice');
Entre_Matrice(n,a);
writeln;writeln('Introduction du vecteur');
Entre_Vecteur(n,b);


for k:=1 to n do x[k]:=0;

for i:=1 to n do

    begin
    s:=a[i,1]*x[1];
    for j:=2 to n do s:=s+a[i,j]*x[j];
    r[i]:=s-b[i];
    end;

d:=r;k:=0;norme:=ProdScal(n,r,r);

while (k<n) and (sqrt(norme)>epsilon) do

     begin
     k:=k+1;
     for i:=1 to n do
         begin
         s:=a[i,1]*d[1];
         for j:=2 to n do s:=s+a[i,j]*d[j];
         t[i]:=s;
    end;
     alpha:=norme/ProdScal(n,t,d);
     for i:=1 to n do
         begin
         r[i]:=r[i]-alpha*t[i];
         x[i]:=x[i]-alpha*d[i];
         end;
     s:=ProdScal(n,r,r);beta:=s/norme;norme:=s;
     for i:=1 to n do d[i]:=r[i]+beta*d[i];
     end;

writeln;writeln('Solution');
EcritVecteur(n,x);
end.
```